Analysis Randomized Complete Block Designs

Randomized complete block designs (RCBD) can, on the surface, be looked upon as consisting of a random factor (blocks) crossed with a fixed factor (treatment factor) and having only a single replicate per cell. In some gross sense, the model would appear as follows:

Response = Blocks Effects + Treatment Effects + Interaction Effects

Because there is only one observation per block/treatment combination no error term is possible. However, recall that the appropriate term for testing a fixed factor in the presence of a random factor is the interaction term.

An Example

A greenhouse consisting of six benches was to be used for an experiment assessing growth among four varieties of house plants. Because light intensity, humidity and temperature varied throughout the greenhouse it was decided that each bench should contain a complete replication of the experiment. Thus, each bench received each variety of potted plant. The change in plant height (cm) after 2 weeks was recorded:

	Varieties			
Bench	1	2	3	4
1	19.8	21.9	16.4	14.7
2	16.7	19.8	15.4	13.5
3	17.7	21.0	14.8	12.8
4	18.2	21.4	15.6	13.7
5	20.3	22.1	16.4	14.6
6	15.5	20.8	14.6	12.9

The SAS code for analyzing the plant growth data collected under a block design is as follows:

```
title 'Randomized Complete Block Design';
title2 'Analysis of plant growth data';
data a;
input bench variety y @@;
cards;
1 1 19.8 1 2 21.9 1 3 16.4 1 4 14.7
2 1 16.7 2 2 19.8 2 3 15.4 2 4 13.5
3 1 17.7 3 2 21.0 3 3 14.8 3 4 12.8
4 1 18.2 4 2 21.4 4 3 15.6 4 4 13.7
5 1 20.3 5 2 22.1 5 3 16.4 5 4 14.6
6 1 15.5 6 2 20.8 6 3 14.6 6 4 12.9
;
proc print data = a;
proc glm data = a;
class bench variety;
model y = bench variety;
means variety / lsd;
run;
```

Analysis of a Randomized Complete Block Design with a Two-way Treatment Structure

Example

A computer company, to test the efficiency of its new programmable calculator, selected size engineers who were proficient in the use of both this calculator and an earlier model and asked them to work out two problems on both calculators. One of the problems was statistical in nature, the other was an engineering problem. The order of the four calculations was randomized independently for each engineer. The length of time (in minutes) required to solve each problem was observed and is presented in the following table:

	Problem			
	Statistical		Engineering	
	Model		Model	
Engineer	New	Earlier	New	Earlier
1	3.1	7.5	2.5	5.1
2	3.8	8.1	2.8	5.3
3	3.0	7.6	2.0	4.9
4	3.4	7.8	2.7	5.5
5	3.3	6.9	2.5	5.4
6	3.6	7.8	2.4	4.8

The SAS code for analyzing these data are as follows:

```
options pageno = 1;
title 'Analysis of a Randomized Complete Block Design';
title2 'with a Twoway Treatment Structure';
data test;
input engineer calculator $ problem $ time @@;
cards;
1 NEW STAT 3.1  1 OLD STAT 7.5  1 NEW ENG 2.5  1 OLD ENG 5.1
2 NEW STAT 3.8  2 OLD STAT 8.1  2 NEW ENG 2.8  2 OLD ENG 5.3
3 NEW STAT 3.0  3 OLD STAT 7.6  3 NEW ENG 2.0  3 OLD ENG 4.9
4 NEW STAT 3.4  4 OLD STAT 7.8  4 NEW ENG 2.7  4 OLD ENG 5.5
5 NEW STAT 3.3  5 OLD STAT 6.9  5 NEW ENG 2.5  5 OLD ENG 5.4
6 NEW STAT 3.6  6 OLD STAT 7.8  6 NEW ENG 2.4  6 OLD ENG 4.8
;
proc print data = test;
run;
title3 'Analysis of Variance for a Twoway Treatment Structure';
proc glm data = test;
class engineer calculator problem;
model time = engineer calculator problem calculator*problem;
means calculator problem / lines lsd;
run;
```